
README

This README follows the main branch, which may not be the currently published version! Use
the docs for the published version of Bamboo.

Bamboo is now maintained by the BEAM Community!
Thank you to thoughtbot for creating and maintaining Bamboo for so long!
Flexible and easy to use email for Elixir.
	Built-in support for popular mail delivery services. Bamboo ships with
adapters for several popular mail delivery services, including Mandrill,
Mailgun, and SendGrid. It's also quite easy to write
your own delivery adapter if your platform isn't yet supported.
	Deliver emails in the background. Most of the time you don't want or need
to wait for the email to send. Bamboo makes it easy with
Mailer.deliver_later.
	A functional approach to mail delivery. Emails are created, manipulated,
and sent using plain functions. This makes composition a breeze and fits
naturally into your existing Elixir app.
	Unit test with ease. Bamboo separates email creation and email delivery
allowing you to test by asserting against email fields without the need for
special functions.
	Dead-simple integration tests. Bamboo provides helper functions to make
integration testing easy and robust.
	View sent emails during development. Bamboo provides a plug you can use
in your router to view sent emails.
	Integrate with Phoenix out of the box. Use Phoenix views and layouts to
make rendering email easy.

See the docs for the most up to date information.
We designed Bamboo to be simple and powerful. If you run into anything that
is less than exceptional, or you just need some help, please open an issue.

  
    
  
  Installation


To install Bamboo, add it to your list of dependencies in mix.exs.
def deps do
  [{:bamboo, "~> 2.3.0"}]
end
You may also use the latest code available from master instead of a
published version in hex:
def deps do
  [{:bamboo, github: "beam-community/bamboo"}]
end
Once you've added Bamboo to your list, update your dependencies by running:
$ mix deps.get

If you are using Elixir < 1.4, also ensure Bamboo is started alongside your
application:
def application do
  [applications: [:bamboo]]
end

  
    
  
  Getting Started


Bamboo separates the tasks of email creation and email sending. To use Bamboo,
you'll need to define one or more email modules (email creation), define a
mailer module (email sending), and provide some configuration.
To create emails, define an email module within your application.
# some/path/within/your/app/email.ex
defmodule MyApp.Email do
  import Bamboo.Email

  def welcome_email do
    new_email(
      to: "john@example.com",
      from: "support@myapp.com",
      subject: "Welcome to the app.",
      html_body: "<strong>Thanks for joining!</strong>",
      text_body: "Thanks for joining!"
    )
  end
end
In addition to the keyword syntax above you can also compose emails using pipes.
To send emails, define a mailer module for your application that uses
Bamboo's mailer.
# some/path/within/your/app/mailer.ex
defmodule MyApp.Mailer do
  use Bamboo.Mailer, otp_app: :my_app
end
Your configuration will need to know your OTP application, your mailer module,
the adapter you are using, and any additional configuration required by the
adapter itself.
# config/config.exs
config :my_app, MyApp.Mailer,
  adapter: Bamboo.MandrillAdapter,
  api_key: "my_api_key"
Bamboo uses Hackney for making requests.
If you want to pass options to Hackney directly, such as controlling
timeouts, you can use the hackney_opts key:
# config/config.exs
config :my_app, MyApp.Mailer,
  adapter: Bamboo.MandrillAdapter,
  api_key: "my_api_key",
  hackney_opts: [
    recv_timeout: :timer.minutes(1),
    connect_timeout: :timer.minutes(1)
  ]
Other adapter-specific configuration may be required. Be sure to check the
adapter's docs.
Now that you have configured Bamboo and defined your modules, you can deliver
email in fitting places within your application.
defmodule MyApp.SomeControllerPerhaps do
  def send_welcome_email do
    Email.welcome_email()   # Create your email
    |> Mailer.deliver_now!() # Send your email
  end
end
Your application is now set up to send email with Bamboo! :tada:

  
    
  
  Using Adapters


An adapter is a set of instructions for how to communicate with a specific
email delivery service. Bamboo ships with support for several popular
services, there are others made available by the
community, or you can use other services by writing a custom adapter.
To use an adapter, declare it in the configuration for your mailer:
# config/config.exs
config :my_app, MyApp.Mailer,
  adapter: Bamboo.MandrillAdapter
Bamboo provides adapters for use in development and testing. To use these
adapters, declare them in the environment configuration.
The local adapter stores emails in memory that can be viewed during
development. Declare its use in your dev environment.
# config/dev.exs
config :my_app, MyApp.Mailer,
  adapter: Bamboo.LocalAdapter
The test adapter sends emails to your running process allowing you to test mail
delivery without emails being sent externally. Declare its use in your test
environment.
# config/test.exs
config :my_app, MyApp.Mailer,
  adapter: Bamboo.TestAdapter
You can create new adapters for any environment by implementing the
Bamboo.Adapter behaviour.

  
    
  
  Delivering Emails in the Background


Often times you don't want to send an email right away because it can block
process completion (e.g. a web request in Phoenix). Bamboo provides a
deliver_later function on your mailers to send emails in the background. It
also provides a Bamboo.DeliverLaterStrategy behaviour that you can
implement to tailor your background email sending.
By default, deliver_later uses Bamboo.TaskSupervisorStrategy. This
strategy sends the email right away, but it does so in the background without
linking to the calling process, so errors in the mailer won't bring down your
app.
You can also create custom strategies by implementing the
Bamboo.DeliverLaterStrategy behaviour. For example, you could create
strategies for adding emails to a background processing queue such as exq or
toniq.

  
    
  
  Composing with Pipes


In addition to creating emails with keyword lists you, can use pipe syntax to
compose emails. This is particularly useful for providing defaults (e.g. from
address, default layout, etc.)
defmodule MyApp.Email do
  import Bamboo.Email
  import Bamboo.Phoenix

  def welcome_email do
    base_email() # Build your default email then customize for welcome
    |> to("foo@bar.com")
    |> subject("Welcome!!!")
    |> put_header("Reply-To", "someone@example.com")
    |> html_body("<strong>Welcome</strong>")
    |> text_body("Welcome")
  end

  defp base_email do
    new_email()
    |> from("myapp@example.com") # Set a default from
    |> put_html_layout({MyApp.LayoutView, "email.html"}) # Set default layout
    |> put_text_layout({MyApp.LayoutView, "email.text"}) # Set default text layout
  end
end

  
    
  
  Handling Recipients


The from, to, cc, and bcc addresses can be a string or a 2 element tuple. What
happens if you try to send to a list of MyApp.Users? Transforming your data
structure each time you send an email would be a pain.
# This stinks. Do you want to do this every time you create a new email?
users = for user <- users do
  {user.name, user.email}
end

new_email(to: users)
Bamboo alleviates this pain by providing the Bamboo.Formatter protocol. By
implementing the protocol for your data structure once, you can pass that
struct directly to Bamboo anywhere it expects an address. See the
Bamboo.Email and Bamboo.Formatter docs for more information and
examples.

  
    
  
  Interceptors


It's possible to configure per Mailer interceptors. Interceptors allow you to
modify or block emails on the fly.
# config/config.exs
config :my_app, MyApp.Mailer,
  adapter: Bamboo.MandrillAdapter,
  interceptors: [MyApp.DenyListInterceptor]
end
An interceptor must implement the Bamboo.Interceptor behaviour. To prevent
email being sent, you can block it with Bamboo.Email.block/1.
# some/path/within/your/app/deny_list_interceptor.ex
defmodule MyApp.DenyListInterceptor do
  @behaviour Bamboo.Interceptor
  @deny_list ["bar@foo.com"]

  def call(email) do
    if email.to in @deny_list do
      Bamboo.Email.block(email)
    else
      email
    end
  end
end

  
    
  
  Using Phoenix Views and Layouts


Phoenix is not required to use Bamboo. But if you want to use Phoenix's views
and layouts to render emails, see bamboo_phoenix and Bamboo.Phoenix.

  
    
  
  Viewing Sent Emails


Bamboo comes with a handy plug for viewing emails sent in development. Now you
don't have to look at the logs to get password resets, confirmation links, etc.
Just open up the sent email viewer and click the link.
See Bamboo.SentEmailViewerPlug.
Here is what it looks like:
[image: Screenshot of BambooSentEmailViewer]

  
    
  
  Mandrill Specific Functionality (tags, merge vars, templates, etc.)


Mandrill offers extra features on top of regular SMTP email like tagging, merge
vars, templates, and scheduling emails to send in the future. See
Bamboo.MandrillHelper.

  
    
  
  SendGrid Specific Functionality (templates, substitution tags, scheduled delivery, etc.)


SendGrid offers extra features on top of regular SMTP email like transactional
templates with substitution tags. See Bamboo.SendGridHelper.

  
    
  
  JSON support


Bamboo comes with JSON support out of the box via the Jason library. To use
it, add :jason to your dependencies:
{:jason, "~> 1.0"}
You can customize it to use another library via the :json_library
configuration:
config :bamboo, :json_library, SomeOtherLib

  
    
  
  Testing


Bamboo separates email creation and email sending. Test email creation by
asserting against the email struct created by your functions. For example,
assuming your welcome email accepts a user recipient, provides the correct from
address, and provides specific text, you might test like this:
defmodule MyApp.EmailTest do
  use ExUnit.Case

  test "welcome email" do
    user = {"Ralph", "ralph@example.com"}

    email = MyApp.Email.welcome_email(user)

    assert email.to == user
    assert email.from == "welcome@myapp.com"
    assert email.html_body =~ "<p>Thanks for joining</p>"
    assert email.text_body =~ "Thanks for joining"
  end
end
Test email sending in integration tests by using the Bamboo.TestAdapter
along with Bamboo.Test. For example, assuming during the registration
process of your app an email is sent to the user welcoming them to the
application, you might test this feature like this:
defmodule MyApp.RegistrationTest do
  use ExUnit.Case
  use Bamboo.Test

  # Remember to use the `Bamboo.TestAdapter` in your test config

  test "after registering, the user gets a welcome email" do
    user = new_user()
    expected_email = MyApp.Email.welcome_email(user.email)

    MyApp.Registration.create(user)

    assert_delivered_email expected_email
  end

  defp new_user do
    # Build a user appropriate to your application
  end
end
See the documentation for Bamboo.Test for more examples and additional
helper functions.

  
    
  
  Available Adapters


Here is a list of adapters that either ship with Bamboo or have been made
available by the community. Feel free to open an issue or a PR if you'd like to
add a new adapter to the list.
	Bamboo.LocalAdapter - Ships with Bamboo. Stores email in memory. Great for local development.
	Bamboo.MailgunAdapter - Ships with Bamboo. Thanks to @princemaple.
	Bamboo.MandrillAdapter - Ships with Bamboo.
	Bamboo.SendGridAdapter - Ships with Bamboo.
	Bamboo.TestAdapter - Ships with Bamboo. Use in your test environment.
	Bamboo.CampaignMonitorAdapter - See jackmarchant/bamboo_campaign_monitor.
	Bamboo.ConfigAdapter - See BinaryNoggin/bamboo_config_adapter declare config at runtime.
	Bamboo.FallbackAdapter - See fuelen/bamboo_fallback. Allows using multiple adapters.
	Bamboo.GmailAdapter - See parkerduckworth/bamboo_gmail.
	Bamboo.MailjetAdapter - See moxide/bamboo_mailjet.
	Bamboo.PostmarkAdapter - See pablo-co/bamboo_postmark.
	Bamboo.SendcloudAdapter - See linjunpop/bamboo_sendcloud.
	Bamboo.SesAdapter - See kalys/bamboo_ses.
	Bamboo.SMTPAdapter - See fewlinesco/bamboo_smtp.
	Bamboo.SparkPostAdapter - See andrewtimberlake/bamboo_sparkpost.


  
    
  
  Contributing


Before opening a pull request, please open an issue first.
Once we've decided how to move forward with a pull request:
$ git clone https://github.com/beam-community/bamboo.git
$ cd bamboo
$ mix deps.get
$ mix test
$ mix format

Once you've made your additions and mix test passes, go ahead and open a PR!
We run the test suite as well as formatter checks on CI. Make sure you are using
the Elixir version defined in the .tool-versions file to have consistent
formatting with what's being run on CI.


  

    
Upgrading to Bamboo 2.0
    

Bamboo 2.0 ships with some breaking changes (hence the major version bump). But
don't worry, if you love Bamboo just as it is, there's an easy upgrade path.
There are two breaking changes:
	Bamboo.Phoenix extracted to bamboo_phoenix
	Bamboo.Mailer.deliver_now/2 and deliver_later/2 don't raise on errors

Let's cover each in turn.

  
    
  
  Breaking change: Bamboo.Phoenix extracted to bamboo_phoenix


Bamboo.Phoenix has been extracted to the bamboo_phoenix library. If you use
Bamboo.Phoenix to render your email templates, add bamboo_phoenix to your
dependencies:
defp deps do
  [
    ...
    {:bamboo, "~> 2.0"},
    {:bamboo_phoenix, "~> 1.0"}
    ...
  ]
end

  
    
  
  Breaking change: deliver_now/2/deliver_later/2 return :ok & :error tuples


Bamboo.Mailer's deliver_now/2 and deliver_later/2 no longer raise errors.
Instead, they now return an {:ok, email} and {:error, error}, where the
error is an exception struct or an error message. If you pass response: true as an argument, the return value will be {:ok, email, response}.
If you prefer seeing code, this is the change in @spec signature for
deliver_now/2:
-  @spec deliver_now(Bamboo.Email.t(), Enum.t()) :: Bamboo.Email.t() | {Bamboo.Email.t(), any}
+  @spec deliver_now(Bamboo.Email.t(), Enum.t()) ::
+          {:ok, Bamboo.Email.t()}
+          | {:ok, Bamboo.Email.t(), any}
+          | {:error, Exception.t() | String.t()}
Note that deliver_later/2 will only return errors that happen prior to
scheduling the delivery of the email. What happens once the delivery is
scheduled depends on what delivery strategy you are using.
Those who want to handle errors on their own can now pattern match on the :ok
and :error tuple responses. If you don't want to handle the errors and like
how Bamboo behaves prior to 2.0, there's a simple upgrade path. 👇

  
    
  
  Simple upgrade path


Bamboo.Mailer comes with deliver_now!/2 and deliver_later!/2. Those two
functions mirror the behavior that deliver_now/2 and deliver_later/2 had
before 2.0.
Hopefully, that makes for a simple upgrade path for those who don't want to
handle the {:ok, email} and {:error, error} tuples. You only need to
change:
	deliver_now/2 to deliver_now!/2, and
	deliver_later/2 to deliver_later!/2

Note that deliver_later!/2 will only raise email validation errors before
scheduling the email delivery. What happens after the delivery is scheduled
depends on the delivery strategy you are using (e.g.
TaskSupervisorStrategy).

  
    
  
  TaskSupervisorStrategy


Regardless of whether you use deliver_later/2 or deliver_later!/2, if you
use the TaskSupervisorStrategy for delivering emails, it will continue to
raise errors when emails fail to be delivered.
If you want control of those errors, you can implement a custom delivery
strategy, to handle errors coming from adapter.deliver.
For now, TaskSupervisorStrategy continues to work as it did prior to 2.0,
so no change is needed here to upgrade to Bamboo 2.0.

  
    
  
  Check with your adapter


Each adapter needs to upgrade to satisfy the new adapter.deliver callback.
Check with your adapter to see if it supports the new ok and error tuple
API before upgrading to Bamboo 2.0.
If you use SendGrid, Mailgun, or Mandrill, your adapter is already updated with
Bamboo 2.0.
That's it! For a full list of changes, please refer to the changelog.
And if you find any issues with this upgrade guide, please let us know by
opening an issue or submitting a pull-request.


  

    
Bamboo.Adapter behaviour
    



      
Behaviour for creating Bamboo adapters
All recipients in the Bamboo.Email struct will be normalized to a two item
tuple of {name, address} when delivered through your mailer. For example,
elem(email.from, 0) would return the name and elem(email.from, 1) would
return the email address.
For more in-depth examples check out the
adapters in Bamboo.

  
    
  
  Example


defmodule Bamboo.CustomAdapter do
  @behaviour Bamboo.Adapter

  def deliver(email, config) do
    deliver_the_email_somehow(email)
  end

  def handle_config(config) do
    # Return the config if nothing special is required
    config

    # Or you could require certain config options
    if Map.get(config, :smtp_username) do
      config
    else
      raise "smtp_username is required in config, got #{inspect(config)}"
    end
  end

  def supports_attachments?, do: true
end

    Bamboo.AdapterHelper - bamboo v2.3.1
    
    

    


  
  

    
Bamboo.AdapterHelper 

Bamboo.Attachment 

Bamboo.DeliverLaterStrategy behaviour
    



      
Behaviour for delivering emails with Bamboo.Mailer.deliver_later/1.
Use this behaviour to create strategies for background email delivery. You
could make a strategy using a GenServer, a background job library or whatever
else you decide.

  
    
  
  Bamboo ships with two strategies:


	Bamboo.TaskSupervisorStrategy
	Bamboo.ImmediateDeliveryStrategy


  
    
  
  Example of setting custom strategies


config :my_app, MyApp.Mailer,
  adapter: Bamboo.MandrillAdapter, # or whatever adapter you want
  deliver_later_strategy: MyCustomStrategy

  
    
  
  Example of creating a custom strategy for delivering later using Task.async


defmodule Bamboo.TaskAsyncStrategy do
  @behaviour Bamboo.DeliverLaterStrategy

  # This is a strategy for delivering later using Task.async
  def deliver_later(adapter, email, config) do
    Task.async fn ->
      # Always call deliver on the adapter so that the email is delivered.
      adapter.deliver(email, config)
    end
  end
end

Bamboo.Email 
    



      
Contains functions for composing emails.
Bamboo separates composing emails from delivering them. This separation makes
emails easy to test and makes things like using a default layout or a default
from address easy to do. This module is for creating emails. To actually send
them, use Bamboo.Mailer.

  
    
  
  Handling email addresses


The from, to, cc, and bcc addresses of a Bamboo.Email can be set to any
data structure for which there is an implementation of the
Bamboo.Formatter protocol or a list of such data
structures. Bamboo includes implementations for some common data structures
or you can create your own. All from, to, cc, and bcc addresses are
normalized internally to a two item tuple of {name, address}. See
Bamboo.Formatter for more info.

  
    
  
  Simplest way to create a new email


defmodule MyApp.Email do
  import Bamboo.Email

  def welcome_email(user) do
    new_email(
      from: "me@app.com",
      to: user,
      subject: "Welcome!",
      text_body: "Welcome to the app",
      html_body: "<strong>Welcome to the app</strong>"
    )
  end
end

  
    
  
  Extracting common parts (default layout, default from address, etc.)


Let's say you want all emails to have the same from address. Here's how you
could do that
defmodule MyApp.Email do
  import Bamboo.Email

  def welcome_email(user) do
    # Since new_email/1 returns a struct you can update it with Kernel.struct!/2
    struct!(base_email(),
      to: user,
      subject: "Welcome!",
      text_body: "Welcome to the app",
      html_body: "<strong>Welcome to the app</strong>"
    )
  end

  def base_email do
    new_email(from: "me@app.com")
  end
end
In addition to keyword lists, Bamboo.Emails can also be built using function pipelines.
defmodule MyApp.Email do
  import Bamboo.Email

  def welcome_email(user) do
    base_email()
    |> to(user)
    |> subject("Welcome!")
    |> text_body("Welcome to the app")
    |> html_body("<strong>Welcome to the app</strong>")
  end
end

Bamboo.Formatter protocol
    



      
Converts data to email addresses.
Implementations of the Bamboo.Formatter protocol convert a given data
structure to a two item tuple of {name, address} or an address string. The
opts argument is a map with the key :type and a value of :from, :to,
:cc, or :bcc. The options argument allows functions to pattern match an
address type and format a given data structure differently for different
types of addresses.

  
    
  
  Simple example


Let's say you have a user struct like this.
defmodule MyApp.User do
  defstruct first_name: nil, last_name: nil, email: nil
end
Bamboo can automatically format this struct if you implement the Bamboo.Formatter
protocol.
defimpl Bamboo.Formatter, for: MyApp.User do
  # Used by `to`, `bcc`, `cc` and `from`
  def format_email_address(user, _opts) do
    fullname = "#{user.first_name} #{user.last_name}"
    {fullname, user.email}
  end
end
Now you can create emails like this, and the user will be formatted correctly
user = %User{first_name: "John", last_name: "Doe", email: "me@example.com"}
Bamboo.Email.new_email(from: user)

  
    
  
  Customize formatting based on from, to, cc or bcc


By pattern matching the opts argument, you can format a given data
structure differently for different types of addresses. For example, if you
want to provide the name of the app when sending email on behalf of a user,
you can format the name for all type: :from addresses.
defimpl Bamboo.Formatter, for: MyApp.User do
  # Include the app name when used in a from address
  def format_email_address(user, %{type: :from}) do
    fullname = "#{user.first_name} #{user.last_name}"
    {fullname <> " (Sent from MyApp)", user.email}
  end

  # Just use the name for all other types
  def format_email_address(user, _opts) do
    fullname = "#{user.first_name} #{user.last_name}"
    {fullname, user.email}
  end
end

Bamboo.ImmediateDeliveryStrategy 
    



      
Strategy for Bamboo.Mailer.deliver_later/1 that sends the email
immediately.
This strategy is used and required by the Bamboo.LocalAdapter and
Bamboo.TestAdapter.

Bamboo.Interceptor behaviour
    



      
Behaviour for creating an Interceptor.
An interceptor allows you to modify or block an email before it is sent. To
block an email, it must be marked as blocked with Bamboo.Email.block/1.

  
    
  
  Example


defmodule Bamboo.DenyListInterceptor do
  @behaviour Bamboo.Interceptor
  @deny_list ["bar@foo.com"]

  def call(email) do
    if email.to in @deny_list do
      Bamboo.Email.block(email)
    else
      email
    end
  end
end

Bamboo.LocalAdapter 
    



      
Stores emails locally. Can be queried to see sent emails.
Use this adapter for storing emails locally instead of sending them. Emails
are stored and can be read from Bamboo.SentEmail. Typically this adapter is
used in the dev environment so emails are not delivered to real email
addresses.
You can use this adapter along with Bamboo.SentEmailViewerPlug to view
emails in the browser.
If you want to open a new browser window for every new email, set the option
open_email_in_browser_url to your preview path.

  
    
  
  Example config


# In config/config.exs, or config/dev.exs, etc.
config :my_app, MyApp.Mailer,
  adapter: Bamboo.LocalAdapter,
  open_email_in_browser_url: "http://localhost:4000/sent_emails" # optional

# Define a Mailer. Maybe in lib/my_app/mailer.ex
defmodule MyApp.Mailer do
  use Bamboo.Mailer, otp_app: :my_app
end

Bamboo.Mailer 
    



      
Functions for delivering emails using adapters and delivery strategies.
Adds deliver_now/1, deliver_now!/1, deliver_later/1 and
deliver_later!/1 functions to the mailer module in which it is used.
Bamboo ships with several adapters. It is also possible
to create your own adapter.
See the "Getting Started" section of the README for an
example of how to set up and configure a mailer for use.

  
    
  
  Example


Creating a Mailer is as simple as defining a module in your application and
using the Bamboo.Mailer.
# some/path/within/your/app/mailer.ex
defmodule MyApp.Mailer do
  use Bamboo.Mailer, otp_app: :my_app
end
The mailer requires some configuration within your application.
# config/config.exs
config :my_app, MyApp.Mailer,
  adapter: Bamboo.MandrillAdapter, # Specify your preferred adapter
  api_key: "my_api_key" # Specify adapter-specific configuration
Also you will want to define an email module for building email structs that
your mailer can send. See [Bamboo.Email] for more information.
# some/path/within/your/app/email.ex
defmodule MyApp.Email do
  import Bamboo.Email

  def welcome_email do
    new_email(
      to: "john@example.com",
      from: "support@myapp.com",
      subject: "Welcome to the app.",
      html_body: "<strong>Thanks for joining!</strong>",
      text_body: "Thanks for joining!"
    )
  end
end
You are now able to send emails with your mailer module where you see fit
within your application.

Bamboo.MailgunAdapter 
    



      
Sends email using Mailgun's API.
Use this adapter to send emails through Mailgun's API. Requires that an API
key and a domain are set in the config.
See Bamboo.MailgunHelper for extra functions that can be used by Bamboo.MailgunAdapter (tagging, merge vars, etc.)

  
    
  
  Example config


# In config/config.exs, or config.prod.exs, etc.
config :my_app, MyApp.Mailer,
  adapter: Bamboo.MailgunAdapter,
  api_key: "my_api_key" # or {:system, "MAILGUN_API_KEY"},
  domain: "your.domain" # or {:system, "MAILGUN_DOMAIN"},
  hackney_opts: [
    recv_timeout: :timer.minutes(1)
  ]

# Define a Mailer. Maybe in lib/my_app/mailer.ex
defmodule MyApp.Mailer do
  use Bamboo.Mailer, otp_app: :my_app
end

  
    
  
  API base URI configuration


Mailgun makes a difference in the API base URL between sender
domains from within the EU and outside.
By default, the base URL is set to https://api.mailgun.net/v3.
To override this globally, you can use the Application environment:
Application.put_env(:bamboo, :mailgun_base_uri, "https://api.eu.mailgun.net/v3")
However, for advanced configurations (for instance, for multi-tenant
setups where you pass in the adapter config when an email is sent),
you might want to specify this on the adapter level:
config :my_app, MyApp.Mailer,
  adapter: Bamboo.MailgunAdapter,
  api_key: "my_api_key",
  domain: "your.domain",
  base_uri: "https://api.eu.mailgun.net/v3"

Functions for using features specific to Mailgun
(e.g. tagging, templates).

Sends email using Mandrill's JSON API.
Use this adapter to send emails through Mandrill's API. Requires that an API
key is set in the config. See Bamboo.MandrillHelper for extra functions that
can be used by Bamboo.MandrillAdapter (tagging, merge vars, etc.)

  
    
  
  Example config


# In config/config.exs, or config/prod.exs, etc.
config :my_app, MyApp.Mailer,
  adapter: Bamboo.MandrillAdapter,
  api_key: "my_api_key",
  hackney_opts: [
    recv_timeout: :timer.minutes(1)
  ]

# Define a Mailer. Maybe in lib/my_app/mailer.ex
defmodule MyApp.Mailer do
  use Bamboo.Mailer, otp_app: :my_app
end

Functions for using features specific to Mandrill (e.g. tagging, merge vars,
templates).

Sends email using SendGrid's JSON API.
Use this adapter to send emails through SendGrid's API. Requires that an API
key is set in the config.
If you would like to add a replyto header to your email, then simply pass it in
using the header property or put_header function like so:
put_header("reply-to", "foo@bar.com")
To set arbitrary email headers, set them in the headers property of the Bamboo.Email struct.
Note that some header names are reserved for use by Sendgrid. See
here for full list.

  
    
  
  Example config


# In config/config.exs, or config.prod.exs, etc.
config :my_app, MyApp.Mailer,
  adapter: Bamboo.SendGridAdapter,
  api_key: "my_api_key",
    # or {:system, "SENDGRID_API_KEY"},
    # or {ModuleName, :method_name, []}
  hackney_opts: [
    recv_timeout: :timer.minutes(1)
  ]

# To enable sandbox mode (e.g. in development or staging environments),
# in config/dev.exs or config/prod.exs etc
config :my_app, MyApp.Mailer, sandbox: true

# Define a Mailer. Maybe in lib/my_app/mailer.ex
defmodule MyApp.Mailer do
  use Bamboo.Mailer, otp_app: :my_app
end

Functions for using features specific to Sendgrid.

  
    
  
  Example


email
|> with_template("80509523-83de-42b6-a2bf-54b7513bd2aa")
|> substitute("%name%", "Jon Snow")
|> substitute("%location%", "Westeros")

